
Visão Geral

Em telas onde existe a necessidade de filtros com muitos critérios deve-se utilizar o componente de Filtro Avançado.
Para utilização do componente de Filtro Avançado em um FreeForm são necessárias alguns passos importantes:

Definir a estrutura de critérios da tela;1.
Definir os eventos inicias para preenchimento de dados;2.
Definir os eventos para critérios preenchidos;3.
Criar EventHandler para manipulação de dados.4.

Definir a estrutura de critérios da tela

Como está especificado no documento do componente de Filtro Avançado a estrutura para agrupamento de informações é a seguinte:

AdvancedFilter
GroupContent

FormInput
Component (DateFieldRange, TextInputRange, DataGrid, etc..)

Componente Descrição

AdvancedFilter Componente de Filtro Avançado.

GroupContent Componente para agrupamento de informações que serão apresentadas.

FormInput Componente responsável por apresentar o label acima do componente inserido à ele.

Component Área onde será inserido o componente homologado conforme listagem apresentada no documento de Filtro Avançado.

A estrutura deverá ser definida utilizando o plugin para cadastro de FreeForm, conforme apresentado em documentação Criando um Form, ou utilizando o Importador
de Protótipos em MXML.

Exemplo de estrutura de um componente de Filtro Avançado que será importado:

<?xml version="1.0" encoding="utf-8"?>

<mx:Application xmlns:mx="http://www.adobe.com/2006/mxml" layout="absolute"

xmlns:ns1="com.datasul.framework.advancedfilter.*"

xmlns:totvs="http://www.totvs.com/2009/framework">

<totvs:AdvancedFilter id="advancedFilter"

title="Cliente"

width="600" height="440">

<totvs:GroupContent id="grupo1" label="Grupo 1 - TextInputRange">

<totvs:FormInput id="campo1" label="Campo 1">

<totvs:TextInputRange id="input1"/>

</totvs:FormInput>

</totvs:GroupContent>

<totvs:GroupContent id="grupo2" label="Grupo 2 - HSlider">

<totvs:FormInput id="campo2" label="Campo 2 - Value">

<mx:HSlider id="input2" thumbCount="1" value="0" minimum="0" maximum="10" snapInterval="1" tickInterval="1"/>

</totvs:FormInput>

<totvs:FormInput id="campo3" label="Campo 3 - Values">

<mx:HSlider id="input3" thumbCount="2" minimum="0" maximum="10" snapInterval="1" tickInterval="1"/>

</totvs:FormInput>

</totvs:GroupContent>

<totvs:GroupContent id="grupo3" label="Grupo 3 - DateFieldRange">

<totvs:FormInput id="campo4" label="Campo 4">

<totvs:DateFieldRange id="input4"/>

</totvs:FormInput>

</totvs:GroupContent>

<totvs:GroupContent id="grupo4" label="Grupo 4 - CheckBox">

<totvs:FormInput id="campo5" label="Campo 5">

<mx:CheckBox id="input5"/>

</totvs:FormInput>

</totvs:GroupContent>

<totvs:GroupContent id="grupo5" label="Grupo 5 - NumericStepper">

<totvs:FormInput id="campo6" label="Campo 5">

<mx:NumericStepper id="input6" minimum="0" maximum="100" value="0" stepSize="10" maxChars="3"/>

</totvs:FormInput>

</totvs:GroupContent>

<totvs:GroupContent id="grupo6" label="Grupo 6 - DataGrid">

<totvs:FormInput id="campo7" label="Campo 7 - Single Row">

<totvs:DataGrid id="list" width="160" selectionMode="singleRow">

</totvs:DataGrid>

</totvs:FormInput>

<totvs:FormInput id="campo8" label="Campo 8 - Multiple Rows">

<totvs:DataGrid id="list2" width="160" selectionMode="multipleRows">

</totvs:DataGrid>

</totvs:FormInput>

</totvs:GroupContent>

</totvs:AdvancedFilter>

</mx:Application>

A tela após importação dos dados e deploy do FreeForm:



Informação

Para que o componente de Filtro Avançado efetue a pesquisa nos critérios informados é necessário informar a propriedade label para o componente
FormInput.

Definir os eventos inicias para preenchimento de dados

Após definida a estrutura de critérios da tela, deve-se mapear os eventos iniciais para preenchimento de dados, conforme necessário.
Para definição dos eventos de um componente pode-se consultar a documentação de Associação de eventos de interface.

Para este exemplo estaremos definindo os seguintes eventos:

Componente Nome Evento

DataGrid input7 creationComplete

DataGrid input8 creationComplete

Definir os eventos para critérios preenchidos

Após definidos os eventos inicias para preenchimento de dados, deve-se mapear os eventos para os critérios que foram preenchidos. Por exemplo, ao clicar sobre o
botão <Aplicar> do componente de Filtro Avançado deve-se efetuar alguma interação com os campos preenchidos.

Para este exemplo estaremos definindo os seguintes eventos:

Componente Nome Evento

AdvancedFilter advancedFilter confirmButtonClick

Criar EventHandler para manipulação de dados

Após definidos os eventos para critérios preenchidos, deve-se criar o EventHandler para manipulação de dados. Para este passo pode-se seguir a documentação
Criação do manipulador de eventos de interface.

Exemplo de manipulação de eventos através de EventHandler:

package com.totvs.md.advancedfilter

{

import com.datasul.framework.advancedfilter.model.IAdvancedFilterEnumeration;

public class DataGridValue implements IAdvancedFilterEnumeration

{

private var _seleceted:Boolean;

private var _value:Object;

public function DataGridValue(value:Object=null, selected:Boolean=false)

{

this._value = value;

this._seleceted = selected;

}

public function get value():Object

{

return this._value;

}

public function set value(value:Object):void

{

this._value = value;

}

public function get selected():Boolean

{

return this._seleceted;



}

public function set selected(value:Boolean):void

{

this._seleceted = value;

}

}

}

package com.totvs.md.advancedfilter {

import com.datasul.framework.advancedfilter.event.AdvancedFilterEvent;

import com.datasul.framework.ui.datagrid.view.DataGrid;

import com.datasul.framework.ui.datagrid.view.SelectColumn;

import com.datasul.framework.ui.msg.model.UIMessage;

import com.datasul.framework.ui.msg.util.MessageUtil;

import com.totvs.framework.md.events.FormEventHandler;

import flash.events.Event;

import mx.collections.ArrayCollection;

import mx.controls.advancedDataGridClasses.AdvancedDataGridColumn;

import mx.utils.ObjectUtil;

public class AdvancedFilterEventHandler extends FormEventHandler {

private var valueDataGrid1:ArrayCollection;

private var valueDataGrid2:ArrayCollection;

public function AdvancedFilterEventHandler() {

super();

var i:int;

var value:DataGridValue;

this.valueDataGrid1 = new ArrayCollection();

for ( i = 0; i < 10; i++) {

value = new DataGridValue("Opção " + i.toString());

this.valueDataGrid1.addItem(value);

}

this.valueDataGrid2 = new ArrayCollection();

for ( i = 0; i < 5; i++) {

value = new DataGridValue("Opção " + i.toString());

this.valueDataGrid2.addItem(value);

}

}

public function onInput7CreationComplete(event:Event):void {

var dataGrid:DataGrid;

var columns:Array;

var selectColumn:SelectColumn;

var datagridColumn:AdvancedDataGridColumn;

selectColumn = new SelectColumn();

selectColumn.dataField = "selected";

datagridColumn = new AdvancedDataGridColumn();

datagridColumn.dataField = "value";

datagridColumn.headerText = "Opções";

columns = new Array();

columns.push(selectColumn);

columns.push(datagridColumn);

dataGrid = event.target as DataGrid;

dataGrid.dataProvider = this.valueDataGrid1;

dataGrid.columns = columns;

}

public function onInput8CreationComplete(event:Event):void {

trace(event);

var dataGrid:DataGrid;

var columns:Array;

var selectColumn:SelectColumn;

var datagridColumn:AdvancedDataGridColumn;

selectColumn = new SelectColumn();

selectColumn.dataField = "selected";

datagridColumn = new AdvancedDataGridColumn();

datagridColumn.dataField = "value";

datagridColumn.headerText = "Opções";

columns = new Array();

columns.push(selectColumn);

columns.push(datagridColumn);

dataGrid = event.target as DataGrid;

dataGrid.dataProvider = this.valueDataGrid2;

dataGrid.columns = columns;

}

public function onAdvFilterConfirmButtonClick(event:Event):void {

if ( event is AdvancedFilterEvent) {

var e:AdvancedFilterEvent = event as AdvancedFilterEvent;

var uiMessage:UIMessage = new UIMessage("Informação", e.toString(), UIMessage.INFO, ObjectUtil.toString(e.filterValue));

MessageUtil.showMessage(uiMessage);

}

}

}

}

Neste exemplo foram adicionados as colunas para cada um dos DataGrids em seus eventos de creationComplete. Para o componente AdvancedFilter foi adicionado
o evento confirmButtonClick que será acionado ao clicar sobre o botão <Aplicar>, onde será apresentada uma mensagem com os campos preenchidos na tela de
Filtro Avançado.

Informação

Pode-se observar que foi criada uma classe chamada DataGridValue, esta classe é necessária para criar os itens que serão adicionados ao dataProvider
do componente de DataGrid.
Em anexo segue código fonte com o Exemplo da aplicação.

Ver Também

